35/replication_package/dofiles/08_MainTables.do

***********************************************************************************
// Replication Files
**********************************************************************************
/*
HOW DO BELIEFS ABOUT THE GENDER WAGE GAP AFFECT THE DEMAND FOR PUBLIC POLICY?
Sonja Settele
AEJ:pol
*/
**********************************************************************************

***********************************************************************************
**** Generate Tables 2-8 in main paper 
**********************************************************************************

***********************************************************************************
// Table 2: Correlates of prior beliefs about gender differences in wages
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear
 
* Keep only incentivized priors 
keep if prior1==1

loc experiments "1 2 3 4 5 6 7 8"

preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""
gen cons=1


    reg prior gender [pweight=pweight], vce(r)
    sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum
	
	reg prior democrat indep otherpol [pweight=pweight], vce(r)
	
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
		
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'

loc ++colnum

	reg prior gender democrat indep otherpol [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
		
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum

	reg prior gender democrat indep otherpol femdem femindep femotherpol [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	sigstar femdem, prec(1)
    estadd loc thisstat11 = "`r(bstar)'": col`colnum'
    estadd loc thisstat12 = "`r(sestar)'": col`colnum'
	sigstar femindep, prec(1)
    estadd loc thisstat14 = "`r(bstar)'": col`colnum'
    estadd loc thisstat15 = "`r(sestar)'": col`colnum'
		
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum
	
qui reg prior gender democrat indep otherpol employee [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	sigstar employee, prec(1)
	estadd loc thisstat17 = "`r(bstar)'": col`colnum'
    estadd loc thisstat18 = "`r(sestar)'": col`colnum'
		
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum
	
	reg prior gender democrat indep otherpol employee fememp [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	sigstar employee, prec(1)
	estadd loc thisstat17 = "`r(bstar)'": col`colnum'
    estadd loc thisstat18 = "`r(sestar)'": col`colnum'
	sigstar fememp, prec(1)
	estadd loc thisstat20 = "`r(bstar)'": col`colnum'
    estadd loc thisstat21 = "`r(sestar)'": col`colnum'
		
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum


qui reg prior gender democrat indep otherpol associatemore femassociate [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	sigstar associatemore, prec(1)
	estadd loc thisstat23 = "`r(bstar)'": col`colnum'
    estadd loc thisstat24 = "`r(sestar)'": col`colnum'
	sigstar femassociate, prec(1)
	estadd loc thisstat26 = "`r(bstar)'": col`colnum'
    estadd loc thisstat27 = "`r(sestar)'": col`colnum'
	
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum


	reg prior gender democrat indep otherpol employee fememp associatemore femassociate [pweight=pweight], vce(r)
	sigstar gender, prec(1)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(1)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(1)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	sigstar employee, prec(1)
	estadd loc thisstat17 = "`r(bstar)'": col`colnum'
    estadd loc thisstat18 = "`r(sestar)'": col`colnum'
	sigstar fememp, prec(1)
	estadd loc thisstat20 = "`r(bstar)'": col`colnum'
    estadd loc thisstat21 = "`r(sestar)'": col`colnum'
	sigstar associatemore, prec(1)
	estadd loc thisstat23 = "`r(bstar)'": col`colnum'
    estadd loc thisstat24 = "`r(sestar)'": col`colnum'
	sigstar femassociate, prec(1)
	estadd loc thisstat26 = "`r(bstar)'": col`colnum'
    estadd loc thisstat27 = "`r(sestar)'": col`colnum'
	
	sigstar _cons, prec(1)
	estadd loc thisstat29 = "`r(bstar)'": col`colnum'
    estadd loc thisstat30 = "`r(sestar)'": col`colnum'
	qui sum prior
    estadd loc thisstat32 = r(N): col`colnum'
	
loc ++colnum
	
   
loc rowlabels " " " "Female" " " " " "Democrat" " " " " "Independent" " " " " "Female x Democrat" " " " " "Female x Independent" " " " " "Employee" " " " " "Female x Employee" " " " " "Associate Degree +" " " " " "Female x Ass. Degree +" " " " " "Constant" " " " " "Observations" "
loc rowstats ""

forval i = 1/32 {
    loc rowstats "`rowstats' thisstat`i'"
}

esttab * using "$output\priorcorrAB.tex", replace cells(none) booktabs nonotes nonum compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("(1)" "(2)" "(3)" "(4)" "(5)" "(6)" "(7)" "(8)") ///
 mgroups("Outcome variable: Incentivized prior belief", pattern(1 0 0 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))



***********************************************************************************
// Table 3: Correlates of perceptions 
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

loc experiments "large problem govmore z_mani_index"

*Keep only control group respondents
keep if rand==0

**z-score prior beliefs:  
foreach var of varlist prior{
egen mean_`var'=mean(`var') 
egen sd_`var'=sd(`var')
replace `var'=(`var'-mean_`var')/sd_`var'
drop mean_`var' sd_`var'
}


*Keep only those with prior beliefs above the 5th and below the 95th percentile of the distribution
sum prior,d
keep if prior > r(p5) & prior < r(p95)


// Build Panels A-C		    		   

preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore


/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experiments' {

	
***Panel A: Main correlates: Dem, Female 
	
    qui reg `choice' i.wave democrat indep otherpol gender [pweight=pweight],r
	
	sigstar democrat, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat6 = "`r(bstar)'": col`colnum'
    estadd loc thisstat7 = "`r(sestar)'": col`colnum'

	
***Panel B: Raw correlation: z-scored prior 
	
    qui reg `choice' i.wave prior [pweight=pweight],r
	
	sigstar prior, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'


***Panel C: Prior and main correlates: Dem, Rep, Female 

    qui reg `choice' prior i.wave democrat indep otherpol gender [pweight=pweight],r
	
	sigstar prior, prec(3)
    estadd loc thisstat14 = "`r(bstar)'": col`colnum'
    estadd loc thisstat15 = "`r(sestar)'": col`colnum'
	
	sigstar democrat, prec(3)
    estadd loc thisstat17 = "`r(bstar)'": col`colnum'
    estadd loc thisstat18= "`r(sestar)'": col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat20 = "`r(bstar)'": col`colnum'
    estadd loc thisstat21 = "`r(sestar)'": col`colnum'
	
	qui sum `choice'  
    estadd loc thisstat22 = r(N): col`colnum'	
	
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"	
	
	}


loc rowlabels " "{\bf Panel A}" " " "Democrat" " " " " "Female" " "  "\hline {\bf Panel B}" " " "Prior (z-scored)" " " "\hline {\bf Panel C}" " " "Prior (z-scored)" " " " " "Democrat" " " " " "Female" " " "\hline Observations" "\hline" "
loc rowstats ""  


loc rowstats ""

forval i = 1/22 {
    loc rowstats "`rowstats' thisstat`i'"
}

esttab * using "$output\tab_correlates_AB_5_95_PanelsA_C.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Gender diff.\\ in wages\\are large}" "\shortstack{Gender diff.\\ in wages\\are a problem}"  ///
  "\shortstack{Government\\should mitigate\\gender wage gap}" "\shortstack{Perception\\Index}", pattern(1 1 1 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear



***********************************************************************************
// Table 4: Correlates of demand for specific policies
***********************************************************************************
  
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

loc experiments "quotaanchor AAanchor legislationanchor transparencyanchor UKtool childcare z_lmpolicy_index"

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore


/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experiments' {

	
***Panel A: Main correlates: Dem, Female 
	
    qui reg `choice' i.wave democrat indep otherpol gender [pweight=pweight],r
	
	sigstar democrat, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat6 = "`r(bstar)'": col`colnum'
    estadd loc thisstat7 = "`r(sestar)'": col`colnum'

	
***Panel B: Raw correlation: z-scored prior 
	
    qui reg `choice' i.wave prior [pweight=pweight],r
	
	sigstar prior, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'


***Panel C: Prior and main correlates: Dem, Rep, Female 

    qui reg `choice' prior i.wave democrat indep otherpol gender [pweight=pweight],r
	
	sigstar prior, prec(3)
    estadd loc thisstat14 = "`r(bstar)'": col`colnum'
    estadd loc thisstat15 = "`r(sestar)'": col`colnum'
	
	sigstar democrat, prec(3)
    estadd loc thisstat17 = "`r(bstar)'": col`colnum'
    estadd loc thisstat18= "`r(sestar)'": col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat20 = "`r(bstar)'": col`colnum'
    estadd loc thisstat21 = "`r(sestar)'": col`colnum'
	
	qui sum `choice'  
    estadd loc thisstat22 = r(N): col`colnum'	
	
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"	
	
	}



loc rowlabels " "{\bf Panel A}" " " "Democrat" " " " " "Female" " "  "\hline {\bf Panel B}" " " "Prior (z-scored)" " " "\hline {\bf Panel C}" " " "Prior (z-scored)" " " " " "Democrat" " " " " "Female" " " "\hline Observations" "\hline" "
loc rowstats ""  


loc rowstats ""

forval i = 1/22 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\tab_correlates_AB_5_95_PanelsD_F.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Introduce\\gender\\quotas}" "\shortstack{Statutory\\affirmative\\action}"  ///
  "\shortstack{Stricter\\equal pay\\legislation}" "\shortstack{Wage transp.\\within\\ companies}" "\shortstack{Introduce\\reporting\\website}" "\shortstack{Increase\\subsidies\\to child care}" ///
   "\shortstack{Policy\\demand\\index}", pattern(1 1 1 1 1 1 1 ) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))


***********************************************************************************
// Table 5: Treatment effect on perceptions and demand for specific policies
***********************************************************************************

// Note: The following code produces one table for Panel A and another table for Panels B-C. Stack manually to obtain final table.


clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

* Drop pure control group
drop if rand==0

// Build Panel A
zscore posterior [aweight=pweight], stub(z)


loc experiments "posterior zposterior large problem govmore z_mani_index"
			   
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""



foreach choice in `experiments' {

   reg `choice' T1 $controls [pweight=pweight], vce(r) 
	
		local n = round(e(N))
			
			sigstar T1, prec(3)
			estadd loc thisstat3 = "`r(bstar)'": col`colnum'
			estadd loc thisstat4 = "`r(sestar)'": col`colnum'
			
			estadd loc thisstat7 = "`n'": col`colnum'
			
			loc ++colnum
}


* Calculate sharpened q-values for columns 3-5
 mat def P = J(5, 1, .)
	reg large T1 $controls [pweight=pweight], vce(r)
    qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg problem T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[2, 1] = r(p)		
	reg govmore T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[3, 1] = r(p)		

	
minq P, q("Q") step(0.001)
	
	estadd loc thisstat5 = "[" + string(Q[1, 1], "%9.3f") +"]" : col3
    estadd loc thisstat5 = "[" +string(Q[2, 1], "%9.3f")+"]": col4
    estadd loc thisstat5 = "[" + string(Q[3, 1], "%9.3f")+"]": col5
	
	
loc rowlabels " "\multicolumn{2}{l}{{\bf Panel A: First Stage}}" " " "T$^{74}$" " " "Sharpened q-value" " " "Observations" "

loc rowstats ""

forval i = 1/7 {
    loc rowstats "`rowstats' thisstat`i'"
}	
		

esttab * using "$output/tab_treatment_main_PanelA.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Post. belief about\\fem. rel. wage\\(0-200)}" "\shortstack{Post. belief about\\fem. rel. wage\\(z-scored)}" "\shortstack{Gender diff.\\ in wages\\are large}" "\shortstack{Gender diff.\\ in wages\\are a problem}"  ///
  "\shortstack{Government\\should mitigate\\gender wage gap}" "\shortstack{Perception\\Index\\((2)-(4))}", pattern(1 1 1 1 1 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))
	
eststo clear


// Build Panels B-C


preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

loc experiments "quotaanchor AAanchor legislationanchor transparencyanchor UKtool childcare z_lmpolicy_index"

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore


/* Statistics */

loc colnum = 1
 
  
foreach choice in `experiments' {

***Panel B	

    reg `choice' T1 $controls [pweight=pweight], vce(r)
	local n = round(e(N))
	
    sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat7 = "`n'": col`colnum'
	

***Panel C	
	
	ivregress 2sls `choice' $controls (zposterior = T1) [pweight=pweight], vce(r) first
	local n = round(e(N))
	
    sigstar zposterior, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat13 = "`n'": col`colnum'
	
	
	loc ++colnum

}

* Calculate sharpened q-values for Panel B, columns 1-6
 mat def P = J(6, 1, .)
	reg quotaanchor T1 $controls [pweight=pweight], vce(r)
    qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg AAanchor T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[2, 1] = r(p)		
	reg legislationanchor T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[3, 1] = r(p)		
	reg transparencyanchor T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[4, 1] = r(p)	
	reg UKtool T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[5, 1] = r(p)
	reg childcare T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[6, 1] = r(p)	

	
minq P, q("Q") step(0.001)
	
	estadd loc thisstat5 = "[" + string(Q[1, 1], "%9.3f") +"]" : col1
    estadd loc thisstat5 = "[" +string(Q[2, 1], "%9.3f")+"]": col2
    estadd loc thisstat5 = "[" + string(Q[3, 1], "%9.3f")+"]": col3
    estadd loc thisstat5 = "["+ string(Q[4, 1], "%9.3f")+"]": col4
    estadd loc thisstat5 = "["+ string(Q[5, 1], "%9.3f")+"]": col5
	estadd loc thisstat5 = "["+ string(Q[6, 1], "%9.3f")+"]": col6 
  
  
loc rowlabels " "\multicolumn{2}{l}{{\bf Panel B: Reduced Form}}" " " "T$^{74}$" " " "Sharpened q-value" " " "Observations" "\hline \multicolumn{2}{l}{{\bf Panel C: 2SLS}}" " " "$\widehat{\text{Posterior belief about}}$" "fem. rel. wage (z-scored)" " " "Observations" "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}

esttab * using "$output\tab_treatment_main_PanelBC.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Introduce\\gender\\quotas}" "\shortstack{Statutory\\affirmative\\action}"  ///
  "\shortstack{Stricter\\equal pay\\legislation}" "\shortstack{Wage transp.\\within\\ companies}" "\shortstack{Introduce\\reporting\\website}" "\shortstack{Increase\\subsidies\\to child care}" ///
   "\shortstack{Policy\\demand\\index}", pattern(1 1 1 1 1 1 1 ) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))


   
***********************************************************************************
// Table 6: Persistence of the treatment effect in obfuscated follow-up survey
***********************************************************************************

// Note: Code produces one table for Panel A and another table for Panel B. Stack manually to obtain final table.

clear all

use "$path\data\SurveyStageIIAB_final.dta"

drop if rand==0


// Build Panel A

loc experimentsII "posteriorII zposteriorII problemII govmoreII problemIIHS problemIILS z_maniII_index2"

			   
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experimentsII' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experimentsII' {

***Panel A	

    reg `choice' T1 $controls [pweight=pweight], vce(r)
	local n = round(e(N))
	
	sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar gender, prec(3)
    estadd loc thisstat7 = "`r(bstar)'": col`colnum'
    estadd loc thisstat8 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat13 = "`n'": col`colnum'
   
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}


// FDR control
 
  mat def P = J(6, 1, .)
	reg posteriorII T1 $controls [pweight=pweight], vce(r)
    qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg zposteriorII T1 $controls [pweight=pweight], vce(r)
    qui test T1 = 0
    mat def P[2, 1] = r(p)	
	reg problemII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[3, 1] = r(p)		
	reg govmoreII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[4, 1] = r(p)	
	reg problemIIHS T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[5, 1] = r(p)		
	reg problemIILS T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[6, 1] = r(p)		
	
minq P, q("Q") step(0.001)
	
	estadd loc thisstat5 = "[" + string(Q[1, 1], "%9.3f") +"]" : col1
    estadd loc thisstat5 = "[" +string(Q[2, 1], "%9.3f")+"]": col2
    estadd loc thisstat5 = "[" + string(Q[3, 1], "%9.3f")+"]": col3
    estadd loc thisstat5 = "["+ string(Q[4, 1], "%9.3f")+"]": col4
    estadd loc thisstat5 = "["+ string(Q[5, 1], "%9.3f")+"]": col5
	estadd loc thisstat5 = "["+ string(Q[6, 1], "%9.3f")+"]": col6


    
loc rowlabels " "{\bf Panel A: Perceptions}" " " "T$^{74}$" " " "Sharpened q-value" " " "Female" " " " " "Democrat" " " " " "Observations" "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}



esttab * using "$output\StageIIABmain_PanelA.tex", replace cells(none) booktabs nonotes compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("\shortstack{Post. belief \\about fem. \\rel. wage\\(0-200)}" "\shortstack{Post. belief\\ about fem.\\ rel. wage\\(z-scored)}" "\shortstack{Gender diff.\\ in wages\\are a \\problem}"  ///
 "\shortstack{Government\\should mitigate \\ gender \\ wage gap}" "\shortstack{Gender diff. in\\wages are a\\problem among\\high-skilled}" ///
 "\shortstack{Gender diff. in\\wages are a\\problem among\\low-skilled}"  "\shortstack{Perception\\Index\\((3)-(6))}") ///
    mgroups("Re-elicited Outcomes" "Newly elicited Outcomes" "Summary Index", pattern(1 0 0 0 1 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear



// Build Panel B

loc experimentsIII "AAanchorII legislationanchorII fampolII antidiscII z_lmpolicyII_specific_index z_lmpolicyII_types_index fairII" 
 
preserve 

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experimentsIII' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experimentsIII' {

***Panel A	

    reg `choice' T1 $controls [pweight=pweight], vce(r)
	local n = round(e(N))
	
    sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar gender, prec(3)
    estadd loc thisstat7 = "`r(bstar)'": col`colnum'
    estadd loc thisstat8 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat13 = "`n'": col`colnum'
   
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}

// FDR control 

mat def P = J(2, 1, .)
	reg AAanchorII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg legislationanchorII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[2, 1] = r(p)
	
minq P, q("Q") step(0.001)
	estadd loc thisstat5 = "["+ string(Q[1, 1], "%9.3f")+"]": col1
	estadd loc thisstat5 = "["+ string(Q[2, 1], "%9.3f")+"]": col2		

 mat def P = J(2, 1, .)
	
	reg fampolII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg antidiscII T1 $controls [pweight=pweight], vce(r)
	qui test T1 = 0
    mat def P[2, 1] = r(p)	
	
minq P, q("Q") step(0.001)
	estadd loc thisstat5 = "["+ string(Q[1, 1], "%9.3f")+"]": col3
	estadd loc thisstat5 = "["+ string(Q[2, 1], "%9.3f")+"]": col4	
	

	
loc rowlabels " "{\bf Panel B: Policy Demand}" " " "T$^{74}$" " " "Sharpened q-value" " " "Female" " " " " "Democrat" " " " " "Observations"  "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\StageIIABmain_PanelB.tex", replace cells(none) booktabs nonotes compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("\shortstack{Statutory\\affirmative\\action}" "\shortstack{Stricter\\equal pay\\legislation}" ///
  "\shortstack{Supportive\\ policy}" "\shortstack{Anti- \\discrimination \\policy}"  ///
  "\shortstack{Policy\\demand\\index\\((1)-(2))}"  "\shortstack{Policy\\demand\\index\\((3)-(4))}" "\shortstack{Women's \\ wages\\are\\ fair}") ///
    mgroups("Re-elicited Outcomes" "Newly elicited Outcomes" "Summary Indices" "Mechanism", pattern(1 0 1 0 1 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))



***********************************************************************************
// Table 7: Heterogeneity in the treatment effect by gender and political orientation
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear


// Generate variable that indicates which posterior belief statistic was elicited
gen postgroup =. 
replace postgroup =1 if wave==1&RAND4== 9
replace postgroup = 2 if wave==1&RAND4==10
replace postgroup= 3 if wave==1&RAND4==11
replace postgroup = 4 if wave==2&RAND4==10
replace postgroup=5 if wave==2&RAND4==11


loc experiments1 "posterior quotaanchor AAanchor legislationanchor transparencyanchor UKtool childcare z_lmpolicy_index"
loc experiments "quotaanchor AAanchor legislationanchor transparencyanchor UKtool childcare z_lmpolicy_index"
  
 
// Set up table  
 
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments1' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

// Panel A: Heterogeneity by Gender

/* Column 1 */

reg posterior T1 T1female $controls i.postgroup [pweight=pweight] if rand!=0, vce(r)
    local n = round(e(N))
	
	sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col1
    estadd loc thisstat4 = "`r(sestar)'": col1
	sigstar T1female, prec(3)
    estadd loc thisstat6 = "`r(bstar)'": col1
    estadd loc thisstat7 = "`r(sestar)'": col1
	test 	T1 + T1female = 0
	estadd loc thisstat8 = string(r(p), "%9.3f"): col1
	
	sigstar gender, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col1
    estadd loc thisstat11 = "`r(sestar)'": col1
	
	estadd loc thisstat13 = "`n'": col1

	
	
/* Columns 2-9 */	
loc colnum = 2
loc colnames ""

foreach choice in `experiments' {

    qui reg `choice' T1 T1female $controls [pweight=pweight] if rand!=0, vce(r)
    local n = round(e(N))
	
	sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar T1female, prec(3)
    estadd loc thisstat6 = "`r(bstar)'": col`colnum'
    estadd loc thisstat7 = "`r(sestar)'": col`colnum'
	test 	T1 + T1female = 0
	estadd loc thisstat8 = string(r(p), "%9.3f"): col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat13 = "`n'": col`colnum'

    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"
	
}

// Panel B: Heterogeneity by political orientation

/* Column 1 */

    reg posterior T1 T1democrat T1indep $controls i.postgroup [pweight=pweight] if rand!=0&otherpol!=1, vce(r)
    local n = round(e(N))
	
	sigstar T1, prec(3)
    estadd loc thisstat16 = "`r(bstar)'": col1
    estadd loc thisstat17 = "`r(sestar)'": col1
	sigstar T1democrat, prec(3)
    estadd loc thisstat19 = "`r(bstar)'": col1
    estadd loc thisstat20 = "`r(sestar)'": col1
	
	test 	T1 + T1democrat = 0
	estadd loc thisstat21 = string(r(p), "%9.3f"): col1
	
	sigstar T1indep, prec(3)
    estadd loc thisstat23 = "`r(bstar)'": col1
    estadd loc thisstat24 = "`r(sestar)'": col1
	test 	T1 + T1indep = 0
	estadd loc thisstat25 = string(r(p), "%9.3f"): col1
	
	sigstar democrat, prec(3)
    estadd loc thisstat27 = "`r(bstar)'": col1
    estadd loc thisstat28 = "`r(sestar)'": col1
	sigstar indep, prec(3)
    estadd loc thisstat30 = "`r(bstar)'": col1
    estadd loc thisstat31 = "`r(sestar)'": col1
	
	estadd loc thisstat33 = "`n'": col1

/* Columns 2-9 */
	
loc colnum = 2
loc colnames ""

foreach choice in `experiments' {


    reg `choice' T1 T1democrat T1indep $controls [pweight=pweight] if rand!=0&otherpol!=1, vce(r)
    local n = round(e(N))
	
	sigstar T1, prec(3)
    estadd loc thisstat16 = "`r(bstar)'": col`colnum'
    estadd loc thisstat17 = "`r(sestar)'": col`colnum'
	sigstar T1democrat, prec(3)
    estadd loc thisstat19 = "`r(bstar)'": col`colnum'
    estadd loc thisstat20 = "`r(sestar)'": col`colnum'
	
	test 	T1 + T1democrat = 0
	estadd loc thisstat21 = string(r(p), "%9.3f"): col`colnum'
	
	sigstar T1indep, prec(3)
    estadd loc thisstat23 = "`r(bstar)'": col`colnum'
    estadd loc thisstat24 = "`r(sestar)'": col`colnum'
	test 	T1 + T1indep = 0
	estadd loc thisstat25 = string(r(p), "%9.3f"): col`colnum'
	
	sigstar democrat, prec(3)
    estadd loc thisstat27 = "`r(bstar)'": col`colnum'
    estadd loc thisstat28 = "`r(sestar)'": col`colnum'
	sigstar indep, prec(3)
    estadd loc thisstat30 = "`r(bstar)'": col`colnum'
    estadd loc thisstat31 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat33 = "`n'": col`colnum'
	
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}

	
loc rowlabels " "\multicolumn{2}{l}{{\bf Panel A: Het. by gender}}" " " "T$^{74}$" " " " " "T$^{74}$ x Female" " " "p-value [T$^{74}$ + T$^{74}$ x female]" " " "Female" " " " " "Observations" "\hline \multicolumn{2}{l}{{\bf Panel B: Het. by pol. orientation}}" " " "T$^{74}$" " " " " "T$^{74}$ x Democrat" " " "p-value [T$^{74}$ + T$^{74}$ x Dem.]" " " "T$^{74}$ x Independent" " " "p-value [T$^{74}$ + T$^{74}$ x Indep.]" " " "Democrat" " " " " "Independent" " " " " "Observations" " 
loc rowstats ""

forval i = 1/33 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\HetTreatment_femdemindAB.tex", replace cells(none) booktabs nonotes compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mtitle("\shortstack{Posterior\\belief about\\fem. rel. wage}"  ///
	"\shortstack{Introduce\\gender\\quotas}" "\shortstack{Statutory\\affirmative\\action}"  ///
  "\shortstack{Stricter\\equal pay\\legislation}" "\shortstack{Wage transp.\\within\\ companies}" "\shortstack{Introduce\\reporting\\website}" "\shortstack{Increase\\subsidies\\to child care}" ///
   "\shortstack{Policy\\demand\\index}" ) ///
   mgroups("First Stage" "Policy Demand", pattern(1 1 0 0 0 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear



***********************************************************************************
// Table 8: Treatment effect on beliefs about underlying factors and fairness perceptions
***********************************************************************************


clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

// Code produces two panels. Stack manually to obtain final table

// Drop pure control group
drop if rand==0

loc experimentsA "ambitious talented interested z_personalreasons_index"
loc experimentsB "discrimination boys society z_extreasons_index womenwages"

	
// Panel A:

	   
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experimentsB' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experimentsA' {


    qui reg `choice' T1 $controls [pweight=pweight] if rand!=0, vce(r) 
	local n = round(e(N))
	
    sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar gender, prec(3)
    estadd loc thisstat7 = "`r(bstar)'": col`colnum'
    estadd loc thisstat8 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'

	estadd loc thisstat13 = "`n'": col`colnum'
	
   
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}


// FDR control
mat def P = J(3, 1, .)

	reg ambitious T1 $controls if rand!=0, vce(r)
	qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg talented T1 $controls if rand!=0, vce(r)
	qui test T1 = 0
    mat def P[2, 1] = r(p)	
	reg interested T1 $controls if rand!=0, vce(r)
	qui test T1 = 0
    mat def P[3, 1] = r(p)	
	
	minq P, q("Q") step(0.001)

    estadd loc thisstat5 = "["+ string(Q[1, 1], "%9.3f")+"]": col1
	estadd loc thisstat5 = "["+ string(Q[2, 1], "%9.3f")+"]": col2
	estadd loc thisstat5 = "["+ string(Q[3, 1], "%9.3f")+"]": col3
	
	
loc rowlabels " "{\bf Panel A}" " " "T$^{74}$" " " "Sharpened q-value" " " "Female" " " " " "Democrat" " " " " "Observations" "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\reasonsmainA.tex", replace cells(none) booktabs nonotes compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("Ambitions" "Talent" "Preferences" "Index") ///
    mgroups("Personal Factors", pattern(1 0 0 0 ) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear



// Panel B:

	   
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experimentsB' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""


foreach choice in `experimentsB' {


    qui reg `choice' T1 $controls [pweight=pweight] if rand!=0, vce(r) 
	local n = round(e(N))
	
    sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar gender, prec(3)
    estadd loc thisstat7 = "`r(bstar)'": col`colnum'
    estadd loc thisstat8 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(3)
    estadd loc thisstat10 = "`r(bstar)'": col`colnum'
    estadd loc thisstat11 = "`r(sestar)'": col`colnum'

	estadd loc thisstat13 = "`n'": col`colnum'
	
   
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}


mat def P = J(3, 1, .)

	reg discrimination T1 $controls if rand!=0, vce(r) 
    qui test T1 = 0
    mat def P[1, 1] = r(p)	
	reg boys T1 $controls if rand!=0, vce(r) 
	qui test T1 = 0
    mat def P[2, 1] = r(p)		
	reg society T1 $controls if rand!=0, vce(r) 
	qui test T1 = 0
    mat def P[3, 1] = r(p)		
	
	
minq P, q("Q") step(0.001)
	
	estadd loc thisstat5 = "[" + string(Q[1, 1], "%9.3f") +"]" : col1
    estadd loc thisstat5 = "[" +string(Q[2, 1], "%9.3f")+"]": col2
    estadd loc thisstat5 = "[" + string(Q[3, 1], "%9.3f")+"]": col3
	
loc rowlabels " "{\bf Panel B}" " " "T$^{74}$" " " "Sharpened q-value" " " "Female" " " " " "Democrat" " " " " "Observations" "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}	


esttab * using "$output\reasonsmainB.tex", replace cells(none) booktabs nonotes compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("Discrimination" "Socialization" "Work-Family"  "Index" "of Women's Wages" ) ///
    mgroups("Impersonal Factors" "Fairness", pattern(1 0 0 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear



***********************************************************************************
// Table 9: The role of beliefs about the effectiveness of policies
***********************************************************************************

*Note: Code produces three separate tables for Panels A,B and C. Stack panels manually to obtain final table

use "$path\data\SurveyStageI_AB_final.dta", clear

// Keep treatment groups of Wave B 
drop if rand==0
keep if wave==2

// Construct indicator for above median perceived effectiveness based on summary index
egen median_eff=median(z_eff_index)
gen abovemedeff=z_eff_index>median_eff 
replace abovemedeff=. if z_eff_index==.
gen T1abovemedeff= T1*abovemedeff


// Build Panel A

loc experiments "effdis effAA effworkfam z_eff_index"
  			   
preserve	

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""

foreach choice in `experiments' {


***Panel A	

    qui reg `choice' T1 $controls [pweight=pweight], vce(r)
    sigstar T1, prec(3)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	sigstar gender, prec(3)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	sigstar democrat, prec(3)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	
	qui sum `choice'
    estadd loc thisstat11 = r(N): col`colnum'

   
    loc ++colnum
    loc colnames "`colnames' `"`: var la `choice''"'"

}

	
loc rowlabels " "{\bf Panel A: Treatment Effect}" "T$^{74}$" " " " " "Female" " " " " "Democrat" " " " " "Observations" " "   "
loc rowstats ""

forval i = 1/11 {
    loc rowstats "`rowstats' thisstat`i'"
}



esttab * using "$output\effectivenessAB_PanelA.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Perceived\\effectiveness\\of anti-disc. \\policies}" "\shortstack{Perceived\\effectiveness \\of affirmative\\ action }"  ///
  "\shortstack{Perceived\\effectiveness\\of work-family\\policies}" "\shortstack{Perceived\\effectiveness\\index\\((1)-(3))}"  ///
  , pattern(1 1 1 1 ) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))
	  
eststo clear


// Build Panel B 

loc experiments "quotaanchor AAanchor legislationanchor UKtool childcare z_lmpolicy_index"
		
		
preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""

loc experiments "posterior large problem govmore z_mani_index"

foreach choice in `experiments' {

   reg `choice' T1 T1abovemedeff abovemedeff $controls [pweight=pweight], vce(r) 
	
		local n = round(e(N))
			
			sigstar T1, prec(3)
			estadd loc thisstat3 = "`r(bstar)'": col`colnum'
			estadd loc thisstat4 = "`r(sestar)'": col`colnum'
			
			sigstar T1abovemedeff, prec(3)
			estadd loc thisstat6 = "`r(bstar)'": col`colnum'
			estadd loc thisstat7 = "`r(sestar)'": col`colnum'
			
			sigstar abovemedeff, prec(3)
			estadd loc thisstat9 = "`r(bstar)'": col`colnum'
			estadd loc thisstat10 = "`r(sestar)'": col`colnum'
			
			test T1 + T1abovemedeff =0
			estadd loc thisstat12 = string(r(p), "%9.3f"): col`colnum'
			
			*estadd loc thisstat13 = "`n'": col`colnum'
			
			loc ++colnum
}


loc rowlabels " "\midline" "{\bf Panel B: First Stage}" "T$^{74}$ (a)" " " " " "T$^{74}$ x $\textbf{1}$ (Perceived" "effectiveness > p50) (b) " " " "$\textbf{1}$ (Perceived" "effectiveness > p50)" " " "p-value [(a) + (b) = 0]"  "
loc rowstats ""

forval i = 1/12 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\TreatmentFirst_hetbyeff_PanelB.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
mgroups("\shortstack{Posterior \\ belief about \\fem. rel. wage}" "\shortstack{Gender diff.\\ in wages\\are large}" "\shortstack{Gender diff.\\ in wages\\are a problem}"  ///
  "\shortstack{Government\\should mitigate\\gender wage gap}" "\shortstack{Perception\\index\\((2)-(4))}", pattern(1 1 1 1 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear


// Build Panel C

loc colnum = 1

loc experiments "quotaanchor AAanchor legislationanchor UKtool childcare z_lmpolicy_index"

preserve

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore
 			   
foreach choice in `experiments' {

    qui reg `choice' T1 T1abovemedeff abovemedeff $controls [pweight=pweight], vce(r)
	
	local n = round(e(N))
	
    sigstar T1, prec(3)
    estadd loc thisstat3 = "`r(bstar)'": col`colnum'
    estadd loc thisstat4 = "`r(sestar)'": col`colnum'
	sigstar T1abovemedeff, prec(3)
    estadd loc thisstat6 = "`r(bstar)'": col`colnum'
    estadd loc thisstat7 = "`r(sestar)'": col`colnum'

	sigstar abovemedeff, prec(3)
    estadd loc thisstat9 = "`r(bstar)'": col`colnum'
    estadd loc thisstat10 = "`r(sestar)'": col`colnum'
	
	test 	T1 + T1abovemedeff = 0
	estadd loc thisstat12 = string(r(p), "%9.3f"): col`colnum'
	
	
	estadd loc thisstat13 = "`n'": col`colnum'

			
	loc ++colnum

}


	
loc rowlabels " " " "{\bf Panel C: Reduced Form}" "T$^{74}$ (a)" " " " " "T$^{74}$ x $\textbf{1}$ (perceived" "effectiveness > p50) (b)" " " "$\textbf{1}$ (perceived" "effectiveness > p50" " " "p-value [(a) + (b) = 0]" "\hline Observations" " "   "
loc rowstats ""

forval i = 1/13 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\Treatmentpolicy_hetbyeff_PanelC.tex", replace cells(none) booktabs nonotes nomtitles compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
	mgroups("\shortstack{Introduce\\gender \\quotas}" "\shortstack{Statutory\\affirmative\\ action}"  ///
  "\shortstack{Stricter\\equal pay\\ legislation}" "\shortstack{Introduce\\reporting \\website}" "\shortstack{Increase\\subsidies\\ to child care}" ///
   "\shortstack{Policy\\demand\\Index}", pattern(1 1 1 1 1 1 ) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))
  
  
eststo clear



***********************************************************************************
// Table 10: Correlates of willingness to pay for additional information
***********************************************************************************

use "$path\data\SurveyStageI_AB_final.dta", clear

// Keep only control group
keep if rand==0

// z-score willingness to pay for info 
foreach var of varlist infopaysupport infopayoppose{
egen mean_`var'=mean(`var') 
egen sd_`var'=sd(`var') 
replace `var'=(`var'-mean_`var')/sd_`var'
drop mean_`var' sd_`var'
}


loc experiments "infopaysupport infopaysupport infopaysupport infopaysupport infopaysupport infopaysupport infopaysupport infopaysupport"

preserve	

clear all
eststo clear
estimates drop _all

set obs 10
qui gen x = 1
qui gen y = 1

loc columns = 0

foreach choice in `experiments' {

    loc ++columns
    qui eststo col`columns': reg x y

}

restore

/* Statistics */

loc colnum = 1
loc colnames ""


foreach var in infopaysupport infopayoppose{

    qui reg `var' democrat indep otherpol [pweight=pweight], vce(r)
	local n = round(e(N))
	
	sigstar democrat, prec(3)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
  
	estadd loc thisstat11 = "`n'": col`colnum'

    loc ++colnum
	
	reg `var' gender [pweight=pweight], vce(r)
	local n = round(e(N))
	
	sigstar gender, prec(3)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat11 = "`n'": col`colnum'
	
	loc ++colnum
	
	
	reg `var' z_lmpolicy_index [pweight=pweight], vce(r)
	local n = round(e(N))
	
	sigstar z_lmpolicy_index, prec(3)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat11 = "`n'": col`colnum'

    loc ++colnum
	
	reg `var' z_lmpolicy_index democrat indep otherpol gender [pweight=pweight], vce(r)
	local n = round(e(N))
   
	sigstar democrat, prec(3)
    estadd loc thisstat2 = "`r(bstar)'": col`colnum'
    estadd loc thisstat3 = "`r(sestar)'": col`colnum'
	
	sigstar gender, prec(3)
    estadd loc thisstat5 = "`r(bstar)'": col`colnum'
    estadd loc thisstat6 = "`r(sestar)'": col`colnum'
	
	sigstar z_lmpolicy_index, prec(3)
    estadd loc thisstat8 = "`r(bstar)'": col`colnum'
    estadd loc thisstat9 = "`r(sestar)'": col`colnum'
	
	estadd loc thisstat11 = "`n'": col`colnum'

    loc ++colnum
	
	
}
     
	
loc rowlabels " " " "Democrat" " " " " "Female" " " " " "Policy Demand (Index)" " " " " "Observations" " "   "
loc rowstats ""

forval i = 1/11 {
    loc rowstats "`rowstats' thisstat`i'"
}


esttab * using "$output\infopay_corr.tex", replace cells(none) booktabs nonotes nonum compress alignment(c) nogap noobs nobaselevels label stats(`rowstats', labels(`rowlabels')) ///
  mtitle("(1)" "(2)" "(3)" "(4)" "(5)" "(6)" "(7)" "(8)") 	///
    mgroups("Willingness to pay for progressive info" "Willingness to pay for traditional info" , pattern(1 0 0 0 1 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))

eststo clear